
Abstract

Biglook is an Object Oriented Scheme library
for constructing GUIs. It uses classes of a CLOS-
like object layer to represent widgets and Scheme
closures to handle events. Combining functional
and object-oriented programming styles yields
an original application programming interface
that advocates a strict separation between the
implementation of the graphical interfaces and
the user-associated commands, enabling compact
source code.

The Biglook implementation separates the
Scheme programming interface and the na-
tive back-end. This permits different ports
for Biglook. The current version uses GTK+

and Swing graphical toolkits, while the pre-
vious release used Tk. It is available at:
http://kaolin.unice.fr/Biglook.

Introduction

We have studied the problem of construct-
ing GUI in functional languages by designing a
widget library for the Scheme programming lan-
guage, called Biglook. In this paper we focus on
how to apply the functional style to GUIs pro-
gramming.

Biglook’s primary use is to implement graph-
ical applications (e.g., xload, editors à la
Emacs, browser̀a la Netscape, programming
tools such askbrowse, our graphical Scheme
code browser). Figure 1 presents two screen

Figure 1: Two Biglook applications: a code
browser on the left, “dock” applications on the
right

shots of Biglook applications:i) kbrowse on the
left and ii) the Biglook dock applications,̀a la
NextStep, on the right. These Biglook applica-
tions are used on a daily basis.

By contrast to previous work, no attempt has
been made to make that library familiar to pro-
grammers used to imperative or purely object ori-
ented programming style. On the contrary, our li-
brary introduces an original application program-
ming interface (API) that benefits from the high
level constructions of the extended Scheme im-
plementation named Bigloo [25] which is open
source and freely available since 1992. The main
Bigloo component is an optimizing compiler that
delivers small and efficient applications for the
UnixTM operating system. Bigloo is able to pro-
duce native code (via C) and JVM bytecode. Cur-
rently Biglook uses GTK+ [20] associated with



the Bigloo C back-end and Swing [29] with the
Bigloo JVM back-end. The previous release of
Biglook [12] used Tk [19].

Bigloo implements an object layer inspired
by CLOS [1]. It is a class-based model where
methods override generic functions and are not
declared inside classes as in Smalltalk [13],
O’Caml [22] or Java [14].

Biglook is implemented as a wrapping layer
on top of native widget libraries (that we name
henceforth theback-end). This software architec-
ture saves the effort of implementing low-level
constructions (pixel switching, clipping, event
handling and so on) allowing to focus on the
Scheme implementation of new features.

When designing the Biglook API, we always
had to decide which model to choose: the func-
tional model or the object model. We think that
these two models are not contradictory but com-
plementary. For instance, if the widget hierarchy
naturally fits a class hierarchy, user call-backs are
naturally implemented by the means of Scheme
closures.

In Section 1 we briefly present the Bigloo sys-
tem emphasizing its module system and its object
layer. This section is required for readers unfa-
miliar with the CLOS model and it also serves as
an introduction tovirtual slots. Virtual slotsare
a new Bigloo construction that is required in or-
der to separate the Biglook API made of classes
and the native back-end. They are presented in
Section 2. In Section 3 we present the Biglook
library. We start showing a simple Biglook ap-
plication and its associated source code. Then,
we detail the Biglook programming principles
(widgets creation, event handling, etc.) motivat-
ing our design orientations by programming lan-
guage considerations. In this section we are con-
ducted to compare the functional programming
style and the object oriented one. In Section 4 we
present the Biglook implementation. At last, in
Section 5 we present a comparison with related
work.

1 Bigloo

Bigloo is an open implementation of the
Scheme programming language. From the be-
ginning, the aim was to propose a realistic and
pragmatic alternative to the strict Scheme [17].
Bigloo does not implement “all” of Scheme; for
example, the execution of tail-recursion may al-
locate memory. On the other hand, Bigloo imple-
ments numerous extensions: support for lexical
and syntactic analysis, pattern matching, an ex-
ception mechanism, a foreign interface, an object
layer and a module language. In this Section, we
present Bigloo’s modules and its object model;
that is, class declarations and generic functions.
Virtual slots, which are heavily used in Biglook,
are presented in Section 2.

1.1 Modules

Bigloo modules have two basic purposes: one
is to allow separate compilation and the second is
to increase the number of errors that can be de-
tected by the compiler. Bigloo modules are sim-
ple and they have been designed with the concern
of an easy implementation.

A module is a compilation unit for Bigloo. It
is represented by one or more files and has the
following syntax:

(module module-name
(import import+)�

(export export+)�

(static static+)�

(library static+)�

)

opt-body

Importclauses are used to import bindings in the
module. In order to import, one just needs to state
the identifier to be imported and its source mod-
ule. Note that a shorthand exists to import all the
bindings of a module.

Export and static clauses play a close role.
They point out to the compiler that the module
implements some bindings and distinguish those
that can be used within other modules (they are
exported) and those that cannot (they are static).
These clauses do not contain identifiers but proto-
types. It is then possible to export variables (mu-



table bindings) or functions (read-only bindings).
Static clauses are optional (the bindings of a mod-
ule which are not referenced in a clause are, by
default, static).

Library clauses enable programs to use Bigloo
libraries. A Bigloo library is merely a collec-
tion of pre-compiled modules. Using a library is
equivalent toimportingall the modules compos-
ing the library. Detailed information on Bigloo
modules may be found in the two papers [25, 26].

1.2 Object layer

In this paper we assume a CLOS-like object
model with single inheritance and single dis-
patch. The object layer implemented in Bigloo
is a restricted version of CLOS [1] inspired, to a
great extent, by MEROON[21].

1.2.1 Class declarations

Classes can be declared static or exported. It
is then possible to make a declaration accessible
from another module or to limit its scope to one
module. The abbreviated syntax of a class decla-
ration is:

(class class-id::super-class-id opt-init opt-slots)

A class can inherit from a single super class.
Classes with no specified super class inherit from
theobject class. The type associated with a sub-
class is a subtype of the type of the super class.

A class may be provided with an initialization
function (opt-init) that is automatically called
each time an instance ofclass-idis created. Ini-
tialization functions accept one argument, the
created instance.

A slot may be typed (with the annotation
::type-id ) and may have a default value
(default option). Here are some possible decla-
rations for the traditionalpoint andpoint-3d:

(module module-points

(export (class point

(point-init)

(x::double (default 0.0))

(y::double (default 0.0)))

(class point-3d::point

(z::double (default 0.0)))))

(define point-init

(let ((count 0))

(lambda (obj::point)

(set! count (+ 1 count))

(print "# of points: " count))))

1.2.2 Instances

When declaring a classcla, Bigloo automat-
ically generates the predicatecla?, an allo-
cator instantiate::cla , an instance cloner
duplicate::cla , accessors (e.g.,cla-x for
a slot x), modifiers (e.g.,cla-x-set! for a
slot x) and an abbreviated special access form,
with-access::cla , to allow accessing and
writing slots simply by using their name. Here
is how to allocate and access an instance of
point-3d :

(let ((p (instantiate::point-3d

(y -3.4)

(x 1.0))))

;; The initialization value of a slot can be omitted
;; from the arguments list if it has a default value;
;; this is the case for the slotz of classpoint-3d.
(with-access::point-3d p (x y z)

(sqrt (+ (sqr x) (sqr y) (sqr z)))))

The instantiate and with-access special
forms are implemented by the means of macros
that statically resolve the keyword parameters
(such asx, y andz). For instance, the above ex-
ample is expanded into:

(let ((p (make-point-3d 1.0 -3.4 0.0)))

(sqrt (+ (sqr (point-3d-x p))

(sqr (point-3d-y p))

(sqr (point-3d-z p)))))

As we can see, since macros are expanded at
compile-time, there is no run-time penalty asso-
ciated with keyword parameters.

1.2.3 Generic functions and methods

Generic function declarations are function decla-
rations annotated by thegeneric keyword. They



can be exported, which means that they can be
used from other modules and that methods can
be added to those functions from other mod-
ules. They can also be static, that is, not acces-
sible from within other modules, hich means that
no other modules can add methods. The CLOS

model fits harmoniously with the traditional func-
tional programming style because a function can
be thought as a generic function overridden with
exactly one method. The syntax to define a
generic function is similar to an ordinary function
definition:

(define-generic (fun::type arg::class ...) opt-
body)

Generic functions must have at least one argu-
ment as this will be used to solve thedynamic
dispatchof methods. This argument is of a type
T and it is impossible to override generic func-
tions with methods whose first argument is not
of a subtype ofT . Methods are declared by the
following syntactic form:

(define-method (fun::type arg::class ...) body)

Methods override generic function definitions.
When a generic function is called, the most spe-
cific applicable method, that is the method de-
fined for the closest dynamic type of the instance,
is dynamically selected. A method may explicitly
invoke the next most specific method overriding
the generic function definition for one of its super
classes (a class has only onedirectsuper class but
severalindirectsuper classes) by the means of the
(call-next-method) form. It calls the method
that should have been used if the current method
had not been defined.

Here is an example of a generic function that
illustrates the use of the Bigloo object layer. We
are presenting a function that prints the value of
the slots of thepoint andpoint-3d instances.
This generic function is namedshow:

(define-generic (show o::point))

Then, the generic function is overridden with a
method for classespoint andpoint-3d.

(define-method (show o::point)

(with-access::point o (x y)

(print "x=" x)

(print "y=" y)))

(define-method (show o::point-3d)

(with-access::point-3d o (z)

(call-next-method)

(print "z=" z)))

Hereafter is an example of a call to theshow
generic function.

(let ((p (instantiate::point-3d

(x 10)

(y 20)

(z 465))))

(show p)) ` x=10 y=20 z=465

2 Virtual slots

Bigloo supports two kind of instance slots:
regular slots that have already been described in
Section 1.2.1, andvirtual slots that enable sev-
eral views of a single data. As we will see in
Section 4.2,virtual slots are at the heart of the
Biglook implementation. They are mandatory to
present Biglook to the user as a class based API.
In particular, wrapping native widgets (such as
GTK+ or Swing) for the Bigloo object model re-
quires virtual slots.

Using virtual slots gives the illusion of ac-
cessing the slots of a class instance but instead,
Scheme functions are called. As we have seen in
Section 1.2.2, the compiler automatically defines
gettersandsettersthat access the various values
embedded in the instances regular slots. Access-
ing virtual slots is syntactically identical to ac-
cessing plain slots, but virtual slots differ in the
following way:

� their getters and setters are not generated by
the compiler. They are defined by the user,
in the class definition, using the class slot
options:get andset.

� they are not allocated into memory.

For instance, let us consider a possible rectan-
gle class implementation. An instance ofrect

is characterized by its origin (x0, y0) and either



its upper right point (x1, y1) or its dimension
(width, height). In the following class defini-
tion, thewidth andheight slots are virtual.

(class rect

x0 y0 x1 y1

(width (get (lambda (o)

(with-access::rect o (x0 x1)

(- x1 x0))))

(set (lambda (o v)

(with-access::rect o (x0 x1)

(set! x1 (+ x0 v))))))

(height (get (lambda (o)

(with-access::rect o (y0 y1)

(- y1 y0))))

(set (lambda (o v)

(with-access::rect o (y0 y1)

(set! y1 (+ y0 v)))))))

Setting thewidth virtual slot (resp. theheight
slot) automatically adjusts thex1 value (resp. the
y1 value) andvice versa. No memory is allocated
for width andheight, as theirvaluesare com-
puted each time they are accessed.

3 The Biglook library

Biglook is an Object Oriented Scheme library
for constructing GUIs. It offers an extensive
set of widgets such as labels, buttons, text edi-
tors, gauges, canvases. Most of the functional-
ity it offers are available through classes rather
than throughad-hocfunctions. For instance, in-
stead of having the classical functionsiconify,
deiconify and window-iconified? for the
window widget, Biglook offers the virtual
slot visible to implement this functional-
ity. Setting thevisible slot enables iconifica-
tion/deiconification. Reading thevisible slot
unveils the window iconification state. As we
will see, this design choice yields a simpler API

and allows usage of introspective techniques for
GUIs programming.

In Section 3.1 we first present a small interface
and discuss how to create the widgets which com-
pose it in Section 3.2. Section 3.4 describes the
notion of containerwidget and placement rules.
Finally, in Section 3.5 we show how to make a
widget reactive to an external event such as a
mouse click.

Throughout this section we justify the choices
we have made when designing the Biglook API.

Figure 2: A simple example

Our reflection on how to create a widget or how
to handle interfaces events are presented in Sec-
tions 3.3 and 3.6.

3.1 A Biglook example

Biglook uses a declarative model for constructing
GUIs. This permits a clear separation between
the code of the interface and the code of the ap-
plication. The construction of an interface starts
by declaring the various widgets which compose
it. Then, the behavior of each widget is speci-
fied independently of its creation by associating
an action (a Scheme closure) with a widget spe-
cific event (key pressed, mouse click, mouse mo-
tion, . . . ).

1:(module example (library biglook))

2:
3:(define awin

4: (instantiate::window

5: (title "A Biglook example")))

6:(define acheck

7: (instantiate::check-button

8: (parent awin)

9: (text "underline")))

10:(define aradio

11: (instantiate::radio

12: (parent awin)

13: (orientation 'horizontal)

14: (border-width 2)

15: (texts '("bold" "italic" "plain"))))

16:(define abutton

17: (instantiate::button

18: (parent awin)

19: (relief 'flat)

20: (text "Quit")))

Figure 3: A simple example, the source code

Figure 2 is a screen shot of a simple Biglook
application. It is made of a window (here named
“A Biglook example”), a check button (the toggle
button underline), a radio button group (bold,
italic, plain), and a plain button (Quit). The



source code of this example is given Figure 3.
From that code, we see that in order to access
Biglook classes and functions, the program uses
a library module clause line1. This program cre-
ates a window (line4), and three widgets (line7,
line 11& 17).

In the sections 3.2 and 3.4 we present how to
create widgets and how to place them in a win-
dow. The Figure 2 interface is inert, that is, no
action is associated with the widgets yet. We will
see in Section 3.5 how actions can be associated
with widgets.

3.2 Widget Creation

The graphical objects (i.e.,widgets) defined
by the Biglook library such as menus, labels or
buttons are represented by Bigloo classes. Each
class defines a set of slots that implement the con-
figuration of the instances. Consequently, tuning
the look of a widget consists in assigning correct
values to its slots. The library offersstandard
default values for each widget but these values
can of course be changed. Generally the cus-
tomization is done at widget creation time. For
instance, the radio group of Figure 3 will be dis-
played horizontally and with a border size of 2
pixels (lines13 and14). A particular aspect of a
widget can be changed by setting a new value to
its corresponding slot. For instance, the expres-
sion

(radio-orientation-set! aradio 'vertical)

changes the orientation of the radio group forcing
a re-display of the whole window. Of course, the
value of this slot can be retrieved by just reading
it:

(print (radio-orientation aradio))

Remember that, as seen in Section 1.2.2, instead
of using the functions created by Bigloo that fetch
and write the value of a slot, one may write an
equivalent program using the Bigloo special form
with-access:

(with-access::radio aradio (orientation)

(set! orientation 'vertical)

(print orientation))

In the rest of this paper, we will use either forms
for accessing class slots.

3.3 Reflection on widget creation

Biglook widget creation supports variable
number of arguments and keyword parameters
(parameters that can be passed in any order be-
cause their actual value is associated with a
name). We have found these features very useful
in order to enable declarative programming for
GUI applications. For instance, a plain Biglook
button is characterized by 20 slots. Some of
them describe the graphical representation (col-
ors, border sizes, . . . ), some others describe the
internal state of the button (widget parent, as-
sociated value, associated text, . . . ). In general,
these numerous slots have default values. When
an instance is created, only slots that have no de-
fault values must be provided. Slots are initial-
ized with their default value unless a user value
is specified. As a consequence, the form that
operates widget construction (e.g., class instan-
tiation) must accept a variable number of argu-
ments. Only some slot values must be provided,
others are optional. In addition, because widget
constructors accept a large number of parameters,
it is convenient tonamethem and to be able to
pass them in any order. This is made possible in
Biglook by theinstantiate:: form. As this
form is implemented using macros that are ex-
panded into calls to the class constructors where
each declared slot is provided with a value, there
is no run-time overhead associated with forms
such asinstantiate::.

Lacking variable number of arguments or key-
words disables declarative programming style for
GUIs because widgets have to be created and, in
a second step, specific attributes have to be pro-
vided. Even overloading and class constructors
do not help. Let’s suppose our window classes
implemented in Java AWT [15] or Swing [29]. To
enable a full declarative style, we should provide
the button class definition with220 constructors
(a constructor for each possible combination of
provided slots). Even for much smaller classes,
this is impractical because, in general, overload-
ing dispatches on types only and several slots can
have the same type. For instance, imagine that we
want to change the graphical appearance of our
window. Instead of using the smallest area large



enough to display the three widgets, we want to
force the width of the window to a specific value.
We can turn the definition of Figure 3, line4 to:

(define awin

(instantiate::window

(title "A Biglook example")

(width 300)))

If we want to specify both width and height, we
can use:

(define awin

(instantiate::window

(title "A Biglook example")

(width 300)

(height 200)))

Languages relying on type overloading cannot
propose these different constructors because the
width and the height of a window are of the same
type.

Languages without overloading nor n-ary
functions traditionally use lists to collect optional
and keyworded arguments. In addition to the run-
time cost imposed by the list constructions, the
called function has to dispatch, at runtime, over
the list to set the parameters values. Furthermore,
such a call cannot be statically typed anymore.

3.4 Containers and widget placement

Biglook uses special sort of widgets to en-
able user customized widget placements: the
container class. A container is a widget that
can embed other widgets. Those widgets are
called thechildrenof the container. For instance,
a window such as the one defined line4 of Fig-
ure 3, is a container, it “contains” the three other
widgets. With the exception of windows, all wid-
gets must be associated with a container in order
to be visible on the screen. To associate a widget
with a container, one have to set itsparent slot
(see lines8, 12 & 18 of Figure 3). Biglook pro-
poses several kind of containers: aligned contain-
ers (such as boxes and windows), grid containers,
note pad containers, paned containers, containers
with scrollbars, etc.

Let us present here two examples of contain-
ers. First, let us consider that we want to modify
the interface of Figure 2. We want the buttons
to be displayed horizontally instead of vertically

Figure 4: An horizontal layout

(see Figure 4). For that, we add a new container
in the window, an horizontalbox:

1:(define awin

2: (instantiate::window

3: (title "A Biglook example")))

4:
5:(define abox

6: (instantiate::box

7: (parent awin)

8: (orientation 'horizontal)))

9:
10:(define acheck

11: (instantiate::check-button

12: (parent abox)

13: (text "underline")))

14:...

For the second example, we combine contain-
ers to design complex interfaces. For instance,
the interface of Figure 5 can be implemented as:

Figure 5: Several containers

1:(define awin

2: (instantiate::window

3: (title "A Biglook example")))

4:
5:(define atab

6: (instantiate::notepad

7: (parent awin)))

8:
9:(define apane

10: (instantiate::paned

11: (orientation 'horizontal)

12: (parent atab)))

13:
14:(define ascroll

15: (instantiate::scroll

16: (parent apane)))



17:(define acheck

18: (instantiate::check-button

19: (parent ascroll)

20: (text "underline")))

21:
22:(define aradio

23: (instantiate::radio

24: (parent apane)

25: (border-width 2)

26: (orientation 'horizontal)

27: (texts '("bold" "italic" "plain"))))

28:
29:(define abutton

30: (instantiate::button

31: (parent awin)

32: (relief 'flat)

33: (text "Quit")))

Note that even if the interfaces of Figures 2 and
5 seem quite different, we only need to modify
the parentslot of theacheck andaradio wid-
gets to embed them in the new containersatab

(line 6), apane (line 10) andascroll (line 15).

3.5 Event Management

Biglook widgets allow the creation of com-
plex GUIs with minimal efforts. In general, when
building such interfaces, one of the main difficul-
ties lies in trying to separate the code of the in-
terface from the rest of the program. Making the
GUI code independent from the rest of the appli-
cation is important because:

� GUIs are often built on a trial-fail basis. It
is hard to conceive an interface ex-nihilo,
and it is generally after using it for a while
that the elements of the GUI find their place.
Keeping the code independent from the rest
of the application allows the development
of prototypes of the interface without nasty
consequences on the other parts of the pro-
gram.

� A given program can have several interfaces
according to the device on which it is run
(e.g. graphical screen, PDA, alphanumeric
terminal). With an independent interface
code, different interfaces can be connected
to the same program.

� The GUI of an application can be con-
structed interactively by an interface builder.
In such a case, it is preferable to keep the
mechanically generated code separate from
hand written parts of the application.

Graphical events (mouse click, key pressed,
window destruction . . . ) can be associated with
widgets by the means of the widgetsevent slot.
This slot must contain an instance of the class
event-handlerwhich is defined as:

(class event-handler

(configure::procedure (default ...))

;; window events
(destroy::procedure (default ...))

...

;; mouse events
(press::procedure (default ...))

(enter::procedure (default ...))

...

;; keyboard events
(key::procedure (default ...))

...)

Each slot of an event-handler is a procedure
called a call-back that accepts one argument.
When a graphical action occurs on a widget, the
associated call-back is invoked passing it anevent
descriptor. Those descriptors are allocated by the
Biglook runtime system. They are instances of
theevent class which is defined as:

(class event

;; the widget which receives the event
(widget::widget read-only)

;; the button number or -1
(button::int read-only)

;; the modifiers list (e.g. shift)
(modifiers::pair-nil read-only)

;; the x position of the mouse
(x::int read-only)

;; the y position of the mouse
(y::int read-only)

;; the character pressed or -1
(char::char read-only)

...)

So, modifying the example of Figure 3 for the
Quit button to be aware of mouse button 1 clicks,
we could write the code as follows:

1:(let* ((p (lambda (e)

2: (if (= (event-button e) 1)

3: (exit))))

4: (evt (instantiate::event-handler

5: (press p))))

6: (with-access::button abutton (event)

7: (set! event evt)))

That is, on line4 we allocateevt, an instance of
theeventhandler class. That event handler is
connected to the button line7. The event handler
only reacts to mousepress events. When such
an event is raised, the call-back line1 is invoked,



its formal parametere being bound to an instance
of theevent class. This function checks the but-
ton number of the raised event (line2). When
the first button is pressed the Biglook application
exits.

It is possible to modify already connected call-
backs. For instance, if we want theQuit button to
emit a sound when it is pressed. we can write:

(with-access::button abutton (event)

(with-access::event-handler event (press)

(let ((olde press))

(set! press

(lambda (e)

(beep) (olde e))))))

Since widgets call-backs are plain Scheme clo-
sures, they can be manipulated as first class ob-
jects, as in this example where new call-backs
capture the values of the old call-backs in order
to reuse them.

3.6 Reflection on event handling

Most modern widget toolkits (with the excep-
tion of Qt [5]) use a call-back framework. That
is, user commands are associated with specific
events (such as mouse click, mouse motion, key-
board inputs,. . . ). When an event is raised, the
user command is invoked. We think that the clo-
sure mechanism is the most simple and efficient
way to implement call-backs even if some alter-
natives exist.

The rest of this section discusses how call-
backs can be implemented depending on the fea-
tures provided by the host language used to im-
plement a graphical toolkit.

3.6.1 Languages that support functions with-
out environment

ISO-C [16] supports global functions but no lo-
cal functions. A C function is always top-level
and may only access its parameters and the set of
global variables. C functions have no definition
environment. However, without an environment,
a call-back is extremely restricted. In particular, a
call-back is likely to access the widget that owns
it. In GTK+ (a C toolkit) when a call-back is as-

sociated with an event, an optional value may be
specified that will be passed to the call-back when
the event is raised. This user value actuallyis the
environment of the call-back. GTK+ mimics clo-
sures with its explicit parameter-passing scheme.
We may notice that the allocation and the man-
agement of the closure environment is in charge
of the client application.

3.6.2 Languages with classes

For languages with classes such as Java, another
strategy can be used. Call-backs may be imple-
mented using class member functions. Member
functions may access the object and the object’s
attributes for which they are invoked. Member
functions look like closures. However, member
functions are not closures because they are asso-
ciated with classes. In other words, all the in-
stances of a class share the implementation of all
their member functions. That is, different call-
back implementations require different class dec-
larations. For instance, if one wants to imple-
ment a button with a call-back printing a plain
message and another one emitting a sound, two
classes have to be defined. These class declara-
tions turn out to be an hindrance to simplicity and
readability. In addition, if several events must be
handled by one widget, this technique turns out
to be impractical because it is not possible to de-
fine a new class for each kind of events the widget
must react to (mouse-1, mouse-2, mouse-3, shift-
mouse-1, ctrl-mouse-1, shift-ctrl-mouse-1, . . . ).

To avoid these extra class definitions, Java
has introduced inner classes. An inner class is
a class defined inside another class; it may be
anonymous. Because in GUI programming inner
classes are used to implement call-backs and as
they are numerous, Java proposes a new syntax
that enables within a single expression, to declare
and to instantiate an inner classe. For instance:

button.addActionListener(new ActionListener() {

public void actionPerformed(ActionEvent e) {

a user code that may reference
current lexical bindings

}

})

The expressionnew ActionListener...

deserves some explanation:1) it declares an
anonymous inner class that2) implements the in-



terfaceActionListener and 3) it creates one
unique instance of that new class that is sent to
the methodaddActionListener of theButton
instance. That is, anonymous inner classes are
the exact Java implementation of closures. It is
worth pointing out that Scheme syntax for clo-
sures is far more compact than the Java one.

3.6.3 Closures

Closures are central to GUI programming be-
cause they are one of the most natural way to
implement call-backs. As we have seen, all call-
back based toolkits offer a mechanism similar to
closures. It can be member functions or anony-
mous Java classes or extra parameter passed to
C functions. However, we have found that solu-
tions of non-functional programming languages
are not as convenient as Scheme’slambda ex-
pression because either the user is responsible
of the construction of the object representing the
closure or extra syntax is introduced.

1:(define tree1

2: (instantiate::tree

3: (parent apane)

4: (root object)

5: (node-label class-name)

6: (node-children class-subclasses)))

7:(define tree2

8: (instantiate::tree

9: (parent apane)

10: (root "/")

11: (node-label (lambda (x) x))

12: (node-children directory->list)))

Figure 6: Two Biglook trees

Not only call-backs are easily implemented by
the means of closures but we have also found that
closures are handy to implement pre-existing data
structure visualization. Consider the screen shot,
Figure 7. It is made of two Biglook trees. A
Biglook tree is a visualization of an existing data
structure. That is, a Biglook tree does not contain
data by itself. It onlyvisualizesan existing struc-
ture. A Biglook tree is defined by three slots:i)
the root of the tree (root), ii) a function that ex-
tracts a string which stands for the label of a node
(node-label), iii) a function that computes the
children list of a node (node-children). The
declaration of the trees of Figure 7 are given Fig-

Figure 7: Two Biglook trees

ure 6, lines2 and8. The first tree (tree1) is a
class tree, its root is the Schemeobject class
denoting the root of the inheritance tree (line4).
The second tree (tree2) is a file tree. Its root
is "/", the root of the file system (line10). To
compute the name of a node representing a class
it is only required to extract the name of that
class (line5). The name of a node represent-
ing a directory is the node itself since the node
is a string (line11). The children list of a class
is computed using the Bigloo library function
class-subclasses (line 6). The children list
of a directory is computed by the library function
directory->list (line 12). As one may no-
tice “hooking” a tree to a data structure is a sim-
ple task. The resulting program is compact. We
think that this compactness is another strength of
Scheme closures.

In general, the Biglook API enables small
source codes for GUIs. On a typical graphical
interface we have found that the Biglook source
is about twice smaller than the same interface im-
plemented in Tcl/Tk and about four times smaller
than the interface implemented in C with GTK+ .

4 Implementing Biglook

In this section, we present the overall Biglook
architecture and implementation. Then, we detail
the role ofvirtual slots.

Figure 8: The Biglook software architecture

4.1 Library Architecture

The Biglook library is implemented on top of a
native back-endtoolkit (currently a GTK+ back-
end and a Swing back-end are available). It takes
advantage of the efficiency of the back-endlow
level operations and it imposes a low memory
overhead (about 15% of additional allocations).
The difference in execution time is marginal. As
a consequence, Biglook can be used to implement
applications using complex graphical interfaces
such as the one presented Figure 1. Interested
readers can find precise performance evaluation
of these implementations in a previous technical
report [12].

Because Biglook makes few assumptions
about the underlying toolkit, re-targeting its im-
plementation to another back-end is generally
possible. To be a potential Biglook back-end, a
toolkit must provide: i) a way to identify a par-
ticular component of the interface. Of course all
the toolkits provide this, even if the representa-
tion used can differ, such as an integer, a string,
a function, and so on. ii) an event manager that
does not hide any events. All the toolkits we have
tried satisfy this criteria (Tk, GTK+ and Swing).
In addition, if a back-end lacks some widgets,
Biglook implements them using primitive wid-
gets. Figure 8 shows the underlying architecture
of the Biglook toolkit.

Actual graphical toolkits generally support
these prerequisites and, as such, can be used as
potential Biglook back-end. We will see in Sec-
tion 4.2 that using virtual slots allows the building
of the rest of the library on this minimal basis.

4.2 Virtual Slots and Biglook

A simple widget is a widget that is directly
mapped into a builtin widget. All simple widgets
are implemented according to the same frame-
work: they inherit from the widget class and
they define user customization options. These
options are implemented using virtual slots. We
present here a possible implementation of the
label class. For the sake of simplicity, we as-
sume that this class extends the widget class
with only one additional slot: the text slot that
specifies the label text.

(class widget::object

(builtin-widget read-only)

...)

(class label::widget

(text

(get

(lambda (o::label)

(with-access::label o (builtin)

(<builtin-label-text> builtin))))

(set

(lambda (o::label v::procedure)

(with-access::label o (builtin)

(<builtin-label-text-set!> builtin

v))))))

Implementing the text slot requires virtual slots.
Its getter and setter functions directly interact
with the back-end toolkit. Virtual slots are used
to establish the connection between Biglook user
point of view of widgets and their native imple-
mentation. Virtual slots are used to provide an
object oriented class-based API to Biglook, inde-
pendent of the back-end.

Now that the slots of a label widget are de-
fined, we must define how such a widget has to be
initialized. The class initialization specific code
is given to the system via the generic function
realize-widget. For each class of the library a
method overrides this generic function and must
call the back-end to create the graphical object
associated with the class. For a label, the method
we need to write is:

(define-method (realize-widget o::label)

(with-access::label o (builtin parent)

(set! builtin

(<builtin-make-label> parent))))

This method creates a builtin label via the low
level <builtin-make-label> function and stores



the result in the builtin slot. This slot creates
the link between the Biglook toolkit and the back-
end.

5 Related work

Many functional languages are connected to
widget libraries especially to the Tk toolkit. Few
of them use object-oriented programming except
in the Scheme world, we can cite mainly STk,
SWL and MrEd.

5.1 Scheme widget libraries

STk [11] is a Scheme interpreter extended with
the Tk graphical toolkit. To some extent, STk is
the ancestor of Biglook, since it was developed
by one of the authors of this paper. However,
STk is tightly coupled to the Tk toolkit and even
if this toolkit is presented to the user through an
object oriented API as in Biglook, no provision
was made to be independent from this back-end.

SWL is a contribution to the Petite Chez
Scheme system [6]. It relies on an interpreter and
uses Tk as back-end. In SWL, native Tk wid-
gets are mapped to Chez Scheme classes and in
this respect this toolkit is similar to the Biglook
or STk libraries. However, SWL implementation
is very different from the one used by those li-
braries since SWL widgets explicitly talk with a
regular Tcl/Tk evaluator.

MrEd [9], a part of the DrScheme project [7],
is a programming environment that contains an
interpreter, a compiler and other various pro-
gramming tools such as browser, debugger, etc.
The back-end toolkit used by MrEd is wxWin-
dow [27], a toolkit that is available under various
platforms (Unix, Windows, etc.). As STk, this
toolkit is completely dependent of its back-end.

5.2 Other functional languages widget li-
braries

Other functional languages provide graphical
primitives using regular functions. The main con-

tribution for strict functional programming lan-
guages has been developed for the Caml pro-
gramming language [30].

The first attempt, CamlTk, is quickly surveyed
in [23]. The design of CamlTk is different from
the one of Biglook. CamlTk bindsTk functions
in Caml while Biglook provides an original API

made of classes.

Recently a new widget library based on GTK+

has been proposed for Caml. No article describes
that connection. However, some work has been
described to add keyword parameters to Caml
in order to help the connection with widget li-
braries [10]. The philosophy of that work dif-
fers from ours because that new library makes the
GTK+ API available from Caml. No attempts
are made to present a neutral API as we did for
Biglook.

Programming graphical user interfaces with
lazy languages is far more challenging than with
strict functional languages. The problem is to
tame the imperative aspects of graphical I/O
in such languages [18, 28]. Several solutions
have been proposed: Fudget by Carlsson and
Hallgren [2, 3], Haggis by Finne and Peyton
Jones [8], TkGofer by Vullings and Claessen [4]
and, more recently the extension of the former
TclHaskell library: FranTk by Sage [24].

Conclusion

In this paper we have presented Biglook, a
widget library for the Bigloo system. The archi-
tecture of Biglook enables different ports. Cur-
rently two ports are available: a GTK+ port and
a Swing port. Biglook source code can be indif-
ferently linked against the two libraries. Biglook
API uses an object oriented programming style
to handle graphical objects and a functional ori-
ented programming style to implement the inter-
face reactivity. We have found that combining
the two programming styles enables more com-
pact implementations for GUIs than most of the
other graphical toolkits.

